

3.87221
49.1

BASIC 语言程序库

自动化工程中常用算法

张巨洪 朱军 刘祖照 李仰东 编
林行良 审

清华大学出版社

内容提要

本书是从自动化工程实际应用角度编制的常用数学程序。共90个。用BASIC语言写成。涉及的方面有：数值计算、信号分析与处理、概率计算与统计、最优化技术以及多种仿真实用程序等。这些程序为各种微处理机、小型机用户开发自己的应用软件提供了方便。

本书在编排中注意由浅入深、联系实际。每个程序都有算法说明、程序说明、操作说明、程序清单及试题上机运行结果。介绍较为完整。各程序相互独立。

本书可供从事计算机和计算机应用的工程技术人员使用，也可作为计算机专业、自动化工程及有关专业、计算数学专业的大专院校师生的参考资料，对于BASIC语言的初学者也是一本有益的参考书。

BASIC语言程序库

自动化工程中常用算法

张巨洪 朱军 刘祖照 李东仰 编
林行良 审

清华大学出版社出版

北京 海淀 清华园

轻工业出版社印刷厂排版

北京岳各庄印刷厂印刷

新华书店北京发行所发行 各地新华书店经售

开本：787×1092 1/16 印张：27 字数：691千字

1983年1月第一版 1983年 月第一次印刷

印数：1—5,000

统一书号：15235·57 定价：3.30元

JS100/18

前 言

在几种常用的计算机算法语言中, BASIC 语言比 ALGOL、FORTRAN 语言简单、易学, 并且目前大量的国产 DJS-100 系列电子计算机和各种微处理机都配有 BASIC 算法语言。

1. 普及推广 BASIC 语言, 使各行各业中大多数人很快地掌握一种使用计算机的本领, 将电子计算机这个二十世纪中出现的最先进的设备和技术在祖国各行各业中充分发挥应有的作用。本书的前六章就是为此而编写的。只要具有初中以上文化水平的人, 稍加学习一些 BASIC 语言的专门教科书和说明书, 再结合本书前六章的练习和上机使用, 就可基本上掌握应用 BASIC 算法语言的能力。

本书的第七章到第十七章, 给出了一般工程上应用的常用算法程序, 包括有定积分、高次方程求根、求解微分方程、线性代数计算、插值、线性回归、快速富里叶变换、滤波、主规划、动态规划等等。这些算法程序在我们工作中经常使用, 对于一般的科学计算、数据处理和工程设计来说, 差不多也都要用到。有了这部分程序, 可使用户节省不少时间和精力。

本书的第十八章到第二十章, 给出了几个控制工程中进行系统设计和系统仿真使用的完整程序。经过本室几年的使用, 这几个程序可以帮助工程技术人员在电子计算机上直接进行控制系统的设计工作, 大大加快工程设计的进程。

在本书的编写过程中, 尽可能使算法说明具体、明确、完整, 每个程序都有试题和完整的上机操作步骤及计算机运行结果。在书的编排中也力争按先简后繁、循序渐进的顺序安排内容。

本书中所有程序都是完整的、独立运行的程序。有的使用者可能要把它们当作一个子程序使用, 只要在程序末尾加一个“RETURN”语句即可。为了不致引起和主程序的矛盾, 本书所有程序都明确地列出程序中所使用的全部数组名和变量名, 便于使用者进行程序连

参加本书编写的还有熊光楞、李芳芸、沈被娜、肖田元以及清华大学自动化系七字班计算机小组的同学。

在本书编写过程中得到了清华大学自动化系系统模拟实验室同志的帮助与支持, 在李仰同志编制程序时, 得到了北京控制工程研究所十室四组同志们的协作与支持。在程序的试编阶段, 也得到了不少教师、研究生和科技工作者的关心和帮助, 在此一并表示感谢。

由于我们水平有限, 一定存在不少错误和不足, 敬请各界读者批评指正。

编 者

1981.6.

32877

目 录

第一章 一般代数运算	(1)
§1. 求一元二次方程的实根和复根	(1)
§2. 双曲函数和反双曲函数	(3)
§3. 二次方程变换 (I)	(5)
§4. 二次方程变换 (II)	(6)
§5. 二次方程变换 (III)	(8)
§6. 求两个整数的最大公因子	(10)
§7. 求一个整数的质因数分解	(11)
§8. 排列和组合	(13)
§9. 求A为底的对数 $\text{LOG}_A(B)$	(14)
第二章 一般几何运算	(16)
§1. 矢量运算	(13)
§2. 矢量分解	(13)
§3. 直角坐标与极坐标的变换	(2)
§4. 直角坐标变换为球坐标	(24)
§5. 角度换算 (I)	(26)
§6. 角度换算 (II)	(27)
§7. 三角多项式的求值	(28)
§8. 平面三角形的解法	(30)
§9. 求平面多边形的面积	(34)
第三章 基本复数运算	(37)
§1. 求复数自变量的三角函数值	(37)
§2. 复数四则运算	(39)
§3. 求一个复数的平方根	(42)
§4. 复数行列式求值	(43)
§5. 列主元消去法解复线性方程组	(47)
§6. 复数矩阵的加减法	(51)
§7. 复数矩阵的乘法	(54)
§8. 求复数矩阵的逆矩阵	(57)
第四章 布尔代数运算	(63)
§1. 求逻辑与、或、异或	(63)
§2. 两个二进制数求和	(66)
§3. 求两个二进制数的商	(70)
§4. 求两个二进制数的积	(75)
§5. 用卡诺图法化简逻辑表达式	(78)

第五章 集合运算	(85)
§1. 求两个集合的并	(85)
§2. 求两个集合的交	(87)
§3. 求两个集合的相对补	(89)
§4. 求两个集合的笛卡尔积	(92)
第六章 概率计算	(95)
§1. 计算实验数据的均值、方差并印出经验分布曲线	(95)
§2. 产生任意区间的均匀分布的随机数	(99)
§3. 产生任意均值和方差的正态分布的随机数	(101)
第七章 特殊函数求值	(105)
§1. 计算贝塞尔函数值	(105)
§2. 计算 I 函数值	(106)
§3. 超几何函数求值	(109)
第八章 高次代数方程求根	(111)
§1. Newton-Raphson 法求多项式的实根	(111)
§2. 半区间搜索法求多项式的实根	(113)
§3. BH 法求多项式的实根和复根	(116)
第九章 求定积分	(123)
§1. 辛普生 (Simpson) 求定积分	(123)
§2. 高斯法求定积分	(124)
§3. 梯形法求定积分	(127)
第十章 微分及龙格-库塔法求解一阶微分方程组	(130)
§1. 微分	(130)
§2. 龙格-库塔法解一阶微分方程组	(132)
第十一章 插值	(137)
§1. 线性内插法	(137)
§2. 抛物线插值法	(138)
§3. 拉格朗日内插法	(140)
§4. 牛顿插值	(142)
§5. 三次样条函数插值	(145)
第十二章 数据平滑与滤波	(152)
§1. 五点三次平滑	(152)
§2. α - β - γ 滤波器	(155)
§3. 离散随机线性系统的 Kalman 滤波器	(171)
>第十三章 信号分析	(189)
§1. 直接法进行富里叶分解	(189)
§2. 直接法求相关函数	(193)
§3. 快速富里叶分解	(196)
§4. 用快速富里叶变换 (FFT) 求相关函数	(211)
第十四章 线性代数计算	(219)

§1. 用高斯-约旦消去法求逆矩阵	(219)
§2. 用高斯-约旦迭代法求逆矩阵	(223)
§3. 矩阵相乘	(226)
§4. 求行列式的值	(229)
§5. 求实对称矩阵特征值和特征向量	(233)
§6. 矩阵相加、相减和标量相乘	(239)
§7. 矩阵求迹、求秩	(243)
§8. 迭代法解线性方程组	(247)
§9. LU分解法解线性方程组	(250)
§10. 高斯——约旦法解线性方程组	(254)
§11. 化一般矩阵为上H阵	(258)
§12. QR法求实上H阵特征值	(262)
§13. 用克雷洛夫方法求矩阵的特征多项式	(272)
>第十五章 多元线性回归分析	(277)
>第十六章 一维最优化方法	(285)
§1. 直接优选法求极值	(285)
§2. 二次插值法求极值	(289)
§3. 有理插值求极值	(293)
>第十七章 线性规划、动态规划	(298)
§1. 非全人工变数单纯形法解线性规划问题	(298)
§2. 一维动态规划	(307)
第十八章 控制系统的计算机辅助设计	(316)
§1. 画控制系统传递函数的NYQIST图	(316)
§2. 由系统的频率响应特性数据拟合系统的传递函数	(323)
第十九章 随机服务系统的仿真与马尔可夫链分析	(337)
§1. 随机服务系统简介	(337)
§2. 单队单服务员的随机服务系统仿真	(337)
§3. 多队单服务员的随机服务系统仿真	(345)
§4. 马尔可夫链分析	(351)
第二十章 控制系统仿真	(365)
§1. 面向微分方程的数字仿真程序	(365)
§2. 连续系统结构图法数字仿真程序	(372)
§3. 连续系统离散相似法数字仿真程序	(384)
§4. 单纯形法寻优程序	(398)
§5. 线性定常系统最佳控制仿真设计程序	(411)
附录: BASIC语言基本语句	(426)

第一章 一般代数运算

§1. 求一元二次方程的实根和复根.

一、方法概要:

设二次方程为 $A_1 X^2 + A_2 X + A_3 = 0$

若判别式 $A_2^2 - 4 A_1 A_3 \geq 0$, 二次方程有两个实根:

$$X_{1,2} = \frac{-A_2 \pm \sqrt{A_2^2 - 4 A_1 A_3}}{2 A_1}$$

若判别式 $A_2^2 - 4 A_1 A_3 < 0$, 二次方程有两个复根:

$$X_{1,2} = \frac{-A_2 \pm I \sqrt{-(A_2^2 - 4 A_1 A_3)}}{2 A_1}$$

若判别式 $A_2^2 - 4 A_1 A_3 = 0$, 二次方程有两个重根.

二、程序说明:

1. 程序使用工作单元 $A_1, A_2, A_3, X_1, X_2, I, D, A, T_1$
2. 程序结束, 二个根的实部在 X_1 和 X_2 中, 虚部在 I_1 和 I_2 中.

三、操作说明

1. 本程序由 BASIC 解释程序引入机内, 由键盘命令 RUN 启动本程序运行, 电传自动印出: "A₁, A₂, A₃" 询问二次方程的各个系数.
2. 由键盘输入二次方程系数 A_1, A_2, A_3 之值, 机器自动印出结果:

$$X_1 = * * * + I * * *$$

$$X_2 = * * * - I * * *$$

当只有两个实根时, 只印出实部, 虚部不印出.

3. 然后机器又印出: "MORE INPUT A?"
询问还有新的二次方程要求根吗? 若有, 由键盘输入 "1", 否则回答其他任意数.
4. 若回答 "1", 机器又印出 "A₁, A₂, A₃" 重复第 (2) 步. 若回答为非 "1" 数, 程序结束.

四、试题:

1. $X^2 + 2X + 3 = 0$

根: $X_1 = -1 + \sqrt{2} I = -1 + 1.41421 I$

$X_2 = -1 - \sqrt{2} I = -1 - 1.41421 I$

2. $X^2 + 2X + 1 = 0$

根: $X_1 = -1$

$X_2 = -1$

3. $12.5X^2 - 15X + 54 = 0$

根: $X_1 = 0.6 + 1.98997 I$

$X_2 = 0.6 - 1.98997 I$

五、程序清单及运行结果

```
7000 PRINT "A1,A2,A3"
7002 INPUT A1,A2,A3
7004 PRINT
7006 LET T1=A2/(2*A1)
7008 LET D=A2*A2-4*A1*A3
7010 IF D<0 THEN GOTO 7020
7012 LET X1=-T1-SQR(D)/(2*A1)
7014 LET X2=-T1+SQR(D)/(2*A1)
7016 PRINT "X1=",X1,"X2=",X2
7018 GOTO 7030
7020 LET X2=-T1
7022 LET X1=-T1
7024 LET I=SQR(-D)/(2*A1)
7026 PRINT "X1=",X1,"+I",ABS(I)
7028 PRINT "X2=",X2,"-I",ABS(I)
7030 PRINT "MORE INPUT A"
7032 INPUT A
7034 PRINT
7036 IF A=1 THEN GOTO 7000
7038 END
```

运行

RUN

A1,A2,A3

? 1? 2? 3

X1=-1+I 1.41421

X2=-1-I 1.41421

MORE INPUT A?

? 1

A1,A2,A3

? 1? 2? 1

X1=-1 X2=-1

MORE INPUT A?

? 1

A1,A2,A3

? 12.5? -15? 54

X1=.6+I 1.98997

X2=.6-I 1.98997

MORE INPUT A?

? 0

§2 双曲函数和反双曲函数

一、方法概要

本程序可用以计算角度自变量为 x (单位: 弧度) 的双曲函数和反双曲函数共十个。

应注意 x 的取值不应使 e^x 溢出。

在本程序中将用到下列计算公式和代码:

$$\text{代码"1"代表双曲正弦} \quad \text{sh}x = \sinh x = \frac{e^x - e^{-x}}{2}$$

$$\text{代码"2"代表双曲余弦} \quad \text{ch}x = \cosh x = \frac{e^x + e^{-x}}{2}$$

$$\text{代码"3"代表双曲正切} \quad \text{th}x = \tanh x = \frac{e^x - e^{-x}}{e^x + e^{-x}}$$

$$\text{代码"4"代表双曲余切} \quad \text{ct}h x = \text{ctgh}x = \frac{e^x + e^{-x}}{e^x - e^{-x}}$$

$$\text{代码"5"代表双曲正割} \quad \text{se}ch x = \frac{1}{\cosh x}$$

$$\text{代码"6"代表双曲余割} \quad \text{cs}ch x = \frac{1}{\sinh x}$$

$$\text{代码"7"代表反双曲正弦} \quad \sinh^{-1}x = \ln(x + \sqrt{x^2 + 1})$$

$$\text{代码"8"代表反双曲余弦} \quad \cosh^{-1}x = \ln(x + \sqrt{x^2 - 1})$$

$$\text{代码"9"代表反双曲正切} \quad \text{tgh}^{-1}x = \frac{1}{2} \ln \frac{1+x}{1-x}$$

$$\text{代码"10"代表反双曲余切} \quad \text{ctgh}^{-1}x = \frac{1}{2} \ln \frac{x+1}{x-1}$$

二、程序说明

程序使用工作单元N, X, V, W

N为所求函数的代码

X为自变量

三、操作说明

1. 由BASIC解释程序将本程序引入机内。键盘命令RUN $\angle$ 启动本程序执行。电传印出:

"CODE, X, TO END PROGRAM INPUT 0, 0?"

2. 键盘输入所求函数的代码N及自变量X的值, 电传自动印出结果并询问下次要求解的函数代码和X值。

3. 若再求新的函数, 则重复2.

若不再计算, 则键盘输入0 $\angle$ 0 $\angle$, 程序自动转到结束。

四、试题:

求 $\text{SINH}(2)$, $\text{COSH}(1)$, $\text{CTANH}(.5)$

ARCSINH(.5).

五、程序清单及运行结果

```
7100 PRINT "CODE,X,TO END PROGRAM INPUT 0,0"
7102 INPUT N,X
7104 PRINT
7106 IF N=0 THEN GOTO 7170
7108 IF N>6 THEN GOTO 7146
7110 LET V=(EXP(X)-EXP(-X))/2
7112 LET W=(EXP(X)+EXP(-X))/2
7114 IF N>4 THEN GOTO 7136
7116 IF N>1 THEN GOTO 7120
7118 PRINT "SINH(",X,")=";V
7119 GOTO 7166
7120 IF N>2 THEN GOTO 7126
7122 PRINT "COSH(",X,")=";W
7124 GOTO 7166
7126 IF N>3 THEN GOTO 7132
7128 PRINT "TANH(",X,")=";V/W
7130 GOTO 7166
7132 PRINT "CTANH(",X,")=";W/V
7134 GOTO 7166
7136 IF N>5 THEN GOTO 7142
7138 PRINT "SECH(",X,")=";1/W
7140 GOTO 7166
7142 PRINT "CSCH(",X,")=";1/V
7144 GOTO 7166
7146 IF N>7 THEN GOTO 7152
7148 PRINT "ARCSINH(",X,")=";LOG(X+SQR(X*X+1))
7150 GOTO 7166
7152 IF N>8 THEN GOTO 7158
7154 PRINT "ARCCOSH(",X,")=";LOG(X+SQR(X*X-1))
7156 GOTO 7166
7158 IF N>9 THEN GOTO 7164
7160 PRINT "ARCTANH(",X,")=";LOG((1+X)/(1-X))/2
7162 GOTO 7166
7164 PRINT "ARCCTANH(",X,")=";LOG((X+1)/(X-1))/2
7166 PRINT "CODE,X"
7168 GOTO 7102
7170 END
```

运行:

4 :

```

RUN
CODE,X,TO END PROGRAM INPUT 0,0
? 1? 2
SINH(2)=3.62686
CODE,X
? 2? 1
COSH(1)=1.54308
CODE,X
? 4? .5
CTANH(.5)=2.16395
CODE,X
? 7? .5
ARCSINH(.5)=.48121
CODE,X
? 0? 0
END AT 7170

```

§3. 二次方程变换(I)

一、算法概要:

本程序是将显形式的二次方程变换成普通的形式。

给定 $Y = PX + Q \pm \sqrt{RX^2 + SX + T}$

要求出: $AX^2 + BXY + CY^2 + DX + EY + F = 0$ 的形式。

其展开方法是将有理项集中在左边, 方程两边平方, 再合并同类项, 以获得要求的系数, 使之成为普通的形式。

$$Y - PX - Q = \pm \sqrt{RX^2 + SX + T}$$

$$(P^2 - R)X^2 + (-2P)XY + Y^2 + (2PQ - S)X + (-2Q)Y + (Q^2 - T) = 0$$

则 $A = P^2 - R$

$$B = -2P$$

$$C = 1$$

$$D = 2PQ - S$$

$$E = -2Q$$

$$F = Q^2 - T$$

二、程序说明

本程序使用工作单元P, Q, R, S, T, A, B, C, D, E, F

计算出结果后, 部分工作单元赋值被破坏

三、操作说明:

1. 由BASIC解释程序将本程序输入机内。
2. 键盘命令RUN启动本程序执行, 电传机询问P, Q, R, S, T之值。
3. 键盘输入P, Q, R, S, T的值, 电传机打印出结果A, B, C, D, E, F的值, 并询问

是否继续作题。

4. 键盘输入“1”则继续作题，重复步骤3，若不再作了：则输入“0”，程序结束。

四、试题：

$$Y=4X+5 \pm \sqrt{6X^2+7X+8}$$

则	A=10	B=-8	C=1
	D=33	E=-10	F=17

五、程序清单及运行结果

```
7300 PRINT "P,Q,R,S,T"
7302 INPUT P,Q,R,S,T
7304 PRINT
7306 PRINT "A=",P*P-R
7308 PRINT "B=", -2*P
7310 PRINT "C=",1
7312 PRINT "D=",2*P*Q-S
7314 PRINT "E=", -2*Q
7316 PRINT "F=",Q*Q-T
7318 PRINT
7320 PRINT "MORE INPUT?(1=YES,0=NO)"
7322 INPUT P
7324 PRINT
7326 IF P=1 THEN GOTO 7300
7328 END
```

运行：

RUN

P, Q, R, S, T

? 4? 5? 6? 7? 8

A=10

B=-8

C=1

D=33

E=-10

F=17

MORE INPUT?(1=YES 0=NO)

? 0

END AT 7328

§4. 二次方程变换 (I)

一、算法概要：

本程序是用来将一个普通的二次方程变换成显形式的二次方程。

给定: $AX^2 + BXY + CY^2 + DX + EY + F = 0$

要求出: $X = PY + Q \pm \sqrt{RY^2 + SY + T}$ 的形式。

由一个二次解所产生的形式除以A, 得到展开式:

$$X = \left(\frac{-B}{2A}\right)Y + \left(\frac{-D}{2A}\right) \pm \sqrt{\left(\left(\frac{B}{2A}\right)^2 - \frac{C}{A}\right)Y^2 + \left(\frac{BD}{2A^2} - \frac{E}{A}\right)Y + \left(\left(\frac{D}{2A}\right)^2 - \left(\frac{F}{A}\right)\right)}$$

这里, 普通形式的系数排列应随所求显形式变量而异。

如: 要求 $X = PY + \dots$ 用 $AX^2 + BXY + \dots$

$Y = PX + \dots$ 则用 $AY^2 + BXY + \dots$

二、程序说明:

程序使用工作单元A, B, C, D, E, F, P, Q, R, S, T.

计算得出结果后, 部分工作单元初始信息破坏。

三、操作说明:

1. 由BASIC解释程序将程序输入机内。
2. 键盘命令RUN启动本程序执行。电传机询问A, B, C, D, E, F值。
3. 键盘输入A, B, C, D, E, F值, 电传机打印出计算结果P, Q, R, S, T值, 并询问是否继续计算。
4. 键盘输入"1", 重新计算, 重复步骤3; 输入"0", 程序结束。

四、试题:

将下面的普通二次方程

$$X^2 + 2XY + 3Y^2 + 4X + 5Y + 6 = 0$$

变换成显形式 $X = PY + Q \pm \sqrt{RY^2 + SY + T}$

计算结果

$$P = -1$$

$$Q = -2$$

$$R = -2$$

$$S = -1$$

$$T = -2$$

即原二次方程变换为

$$X = -Y - 2 \pm \sqrt{-2Y^2 - Y - 2}$$

五、程序清单及运行结果

```
7400 PRINT "ENTER A,B,C,D,E,F"
7402 INPUT A,B,C,D,E,F
7404 PRINT
7406 PRINT "P=", -B/A/2
7408 PRINT "Q=", -D/A/2
7410 PRINT "R=", (B/A/2)*(B/A/2)-C/A
7412 PRINT "S=", (B*D/2/A-E)/A
7414 PRINT "T=", (D/A/2)*(D/A/2)-F/A
7416 PRINT
7418 PRINT "MORE INPUT? (1=YES,0=NO)"
```

```

7420 INPUT A
7422 PRINT
7424 IF A=1 THEN GOTO 7400
7426 END

```

运行

RUN

ENTER A,B,C,D,E,F

? 1? 2? 3? 4? 5? 6

P=-1

O=-2

R=-2

S=-1

T=-2

MORE INPUT?(1=YES,0=NO)

? 1

ENTER A,B,C,D,E,F

? 5? 6? 2? 4? 9? 8

R=-.6

Q=-.4

R=-4.00001E-2

S=.6

T=1.44

MORE INPUT?(1=YES,0=NO)

? 0

END AT 7426

§5. 二次方程变换 (II)

一、算法概要:

本程序是将显形的二次方程 $Y=f(x)$ 变换成 $X=g(y)$ 的形式

给定: $Y=PX+Q \pm \sqrt{RX^2+SX+T}$

要求出 $X=P_1Y+Q_1 \pm \sqrt{R_1Y^2+S_1Y+T_1}$ 的形式。

方法是先将原始方程变换成 $Y-PX-Q=\pm \sqrt{RX^2+SX+T}$, 等式两边平方后变换成普通形式, 然后计算出所求函数的系数。

二、程序说明:

程序使用工作单元P, Q, R, S, T, A, D, P₁, Q₁, R₁, S₁, T₁。

计算得出结果后, 部分工作单元初始信息破坏。

三、操作说明:

1. 由BASIC解释程序将程序输入机内。

2. 键盘命令RUN启动本程序执行。电传机询问P, Q, R, S, T的值。

3. 键盘输入P, Q, R, S, T值后, 电传机打印出结果 P_1 , Q_1 , R_1 , S_1 和 T_1 值, 并询问是否继续作题。

4. 欲继续作题, 键盘输入"1", 如不再作, 键盘输入"0", 程序结束。

四、试题:

$Y=2X+5 \pm \sqrt{8X^2+9X+4}$, 变换成一个Y函数,

即 $X=P_1Y+Q_1 \pm \sqrt{R_1Y^2+S_1Y+T_1}$ 形式。

计算结果

$$P_1 = \frac{P}{A} = -0.5$$

$$Q_1 = -\frac{D}{2A} = 1.375$$

$$R_1 = \left(\frac{P}{A}\right)^2 - \frac{1}{A} = 0.5$$

$$S_1 = -DP/A^2 + 2Q/A = -3.875$$

$$T_1 = \left(\frac{D}{2A}\right)^2 - (Q^2 - T)/A = 7.141$$

其中 $A=P*P-R$, $D=2*P*Q-S$

五、程序清单及运行结果

```
7400 PRINT "ENTER P,Q,R,S,T"
7404 INPUT P,Q,R,S,T
7406 PRINT
7408 LET A=P*P-R
7410 LET D=2*P*Q-S
7412 PRINT "P1=",P/A
7414 PRINT "Q1=", -D/2/A
7416 PRINT "R1=", (P/A)*(P/A)-1/A
7418 PRINT "S1=", -D*P/(A*A)+2*Q/A
7420 PRINT "T1=", (D/2/A)*(D/2/A)-(Q*Q-T)/A
7422 PRINT
7424 PRINT "MORE INPUT?(1=YES,0=NO)"
7426 INPUT P
7428 PRINT
7430 IF P=1 THEN GOTO 7400
7432 END
```

运行:

RUN

ENTER P,Q,R,S,T

? 2? 5? 8? 9? 4

P1=-.5

Q1=1.375

R1=.5

```

S1=-3.875
T1=7.14062
MORE INPUT,(I=YES,0=NO)
? 0
END AT 7432

```

§6 求两个整数的最大公因子

一、方法概要

本程序用欧几里德算法计算两个整数的最大公因子。

x, y 为两个整数, 令

$$C=|x|, \quad D=|y|$$

则 $R=C-D*(C/D \text{ 的整数部分})$ (1-6-1)

若 $R=0$, 则 D 即为 x 和 y 的最大公因子。若 $R \neq 0$, 再令 $C=D, D=R$, 重新按公式(1-6-1) 计算出新的 R 值, 周而复始, 直至 $R=0$ 找到最大公因子。

二、程序说明

本程序使用工作单元 A、B、C、D、R。运算结果存在 D 中。

三、操作说明

1. 由 BASIC 解释程序将本程序输入机内。
2. 键盘命令 RUN 启动本程序执行, 电传机询问两个整数的值?
3. 键盘输入两个整数, 电传机立即打出结果:

G·C·D=

并询问另两个整数。

4. 如继续计算, 则重复步骤 3, 如不再计算, 则输入 0, 程序结束。

四、试题:

求 45 和 285 的最大公因子。

计算结果 $G \cdot C \cdot D = 15$

五、程序清单和运行结果

```

7500 PRINT "GREATEST COMMON DIVISOR OF TWO INTEGERS"
7502 PRINT
7504 PRINT "INPUT INTEGER, INTEGER TO END PROGRAM INPUT 0,0"
7506 INPUT A,B
7508 PRINT
7510 IF A=0 THEN GOTO 7536
7512 PRINT
7514 LET C=ABS(A)
7516 LET D=ABS(B)
7518 LET R=C-D*INT(C/D)
7520 IF R=0 THEN GOTO 7528
7522 LET C=D

```

```

7524 LET D=R
7526 GOTO 7518
7528 PRINT "G.C.D=", D
7530 PRINT
7532 PRINT "INPUT INTEGER, INTEGER"
7534 GOTO 7506
7536 END
运行:
RUN
GREATEST COMMON DIVTSOR OF TWO INTEGERS
INPUT INTEGER, INTEGER, TO END PROGRAM INPUT 0,0
? 45? 285
G.C.D=15
INPUT INTEGER, INTEGER
? 57? 33
G.C.D= 3
INPUT INTEGER, INTEGER
? 57? 11
G.C.D=1
INPUT INTEGER, INTEGER
? 0? 0
END AT 7536

```

§7 求一个整数的质因数分解

一、程序说明:

1. 本程序用来将一个整数分解成质数的乘积, 使该整数

$$N=(A_1)^{B_1} * (A_2)^{B_2} * \dots * (A_N)^{B_N}$$

2. 程序占用工作单元N, I, S。

二、操作说明:

1. 由BASIC解释程序将本程序输入机内
2. 键盘命令RUN启动本程序执行, 电传机询问所要分解的整数N值
3. 键盘输入N值, 电传机打印出计算结果:

1 (或-1, 表示数的正负号)

A_1 ↑ B_1 (↑表示乘幂运算符)

A_2 ↑ B_2

⋮ ⋮ ⋮

A_N ↑ B_N

当输入数为0, 1或非整数时, 不予回答。